

Piano per 3 punti e proiezione

Enrico Bertolazzi
Esercitazioni Maple di Geometria

> # procedura per la costruzione di un piano per 3 punti

```
piano_3p := proc(p1,p2,p3)
  local a, b, c, d, plane, eq1, eq2, eq3, sol, lc ;
  # definisce l'equazione generica di un piano come una espressione
  plane := a * x + b * y + c * z + d ;
  # calcolca la espressione nei punti di passaggio
  eq1 := subs(x=p1[1],y=p1[2],z=p1[3],plane) ;
  eq2 := subs(x=p2[1],y=p2[2],z=p2[3],plane) ;
  eq3 := subs(x=p3[1],y=p3[2],z=p3[3],plane) ;
  # ugualinato a zero eq1, eq2, eq3 determino le indeterminate
  # a, b, c, d
  sol := solve({eq1=0,eq2=0,eq3=0},{a,b,c,d}) ;
  # sol contiene le soluzioni a=..., b=...,
  # posso usare la soluzione per effettuare la sostituzione
  plane := subs(sol,plane) ;
  # poiche la soluzione non e' unica ma dipende da uno scalare
  # divido la soluzione per il primo coefficiente non zero
  # del polinomio in x,y,z.
  lc := tcoeff(plane,[x,y,z]) ;
  return simplify(plane/lc) ;
end proc ;
```

piano_3p := proc(p1, p2, p3)

(1)

```
  local a, b, c, d, plane, eq1, eq2, eq3, sol, lc;
  plane := a*x + b*y + c*z + d;
  eq1 := subs(x = p1[1], y = p1[2], z = p1[3], plane);
  eq2 := subs(x = p2[1], y = p2[2], z = p2[3], plane);
  eq3 := subs(x = p3[1], y = p3[2], z = p3[3], plane);
  sol := solve({eq1 = 0, eq2 = 0, eq3 = 0}, {d, a, c, b});
  plane := subs(sol, plane);
  lc := tcoeff(plane, [x, y, z]);
  return simplify(plane/lc)
```

end proc

> # prendo 3 punti nello spazio

```
Q1 := [1,2,3] ;
Q2 := [4,4,1] ;
```

```

Q3 := [1,1,0] ;
                                Q1 := [1, 2, 3]
                                Q2 := [4, 4, 1]
                                Q3 := [1, 1, 0]

```

(2)

```

> # esempio d'uso
piano_3p(Q1,Q2,Q3) ;
                                8x - 9y + 3z + 1

```

(3)

```

> # proiezione punto su un piano (scritto in forma cartesiana)

proj_piano := proc(piano,Q)
  local n, r, eq, sol ;
  # estrae i coefficienti della normale al piano
  n := [ coeff(piano,x,1), coeff(piano,y,1), coeff(piano,z,1) ] ;
  # costruisce la retta parametrica passante per Q e direzione n
  r := expand(Q + t * n) ;
  # costruisce l'equazione per intersezione piano retta
  eq := subs(x=r[1],y=r[2],z=r[3],piano) ;
  # risolve il sistema (1 equazione 1 incognita)
  sol := solve(eq=0,{t}) ;
  # sostituisce la soluzione nella equazione della retta
  # per trovare in punto di proiezione
  return subs(sol,r) ;
end proc ;

```

(4)

```

proj_piano := proc(piano, Q)
  local n, r, eq, sol;
  n := [coeff(piano, x, 1), coeff(piano, y, 1), coeff(piano, z, 1)];
  r := expand(Q + t*n);
  eq := subs(x=r[1],y=r[2],z=r[3],piano);
  sol := solve(eq = 0, {t});
  return subs(sol, r)
end proc

```

end proc

```

> # esempio d'uso:
# costruisce il piano per i punti Q1, Q2, Q3
pl := piano_3p(Q1,Q2,Q3) ;
                                pl := 8x - 9y + 3z + 1

```

(5)

```

> # prendo un punto qualunque e lo proietto sul piano
proj_piano(pl, [1,34,45]) ;
                                [ 725/77, 1889/77, 3708/77 ]

```

(6)